Deriving a general commutation relation for X,y.
Here y,b,a,x are the generators with relations
ax=x(a-α) ay=y(a+α)
by=y(b-β) bx=x(b+β)
xy=yx+(B-A)/$\hbar$
where A = $e^{-\hbar\beta a}$, B = $e^{-\hbar\alpha b}$ and q = $e^{-\alpha\beta\hbar}$
AB = t is a central element. Since we want to expand in β,
we change variables to X = Axt^{-1} $\hbar$ to get
Xy = q yX + 1-G where G = $A^2\,t^{-1}$, so X G = $q^2\,X\,G$ and G y = $q^2\,y\,G$.
also set α=1.

Below we verify the following fundamental commutation relation:

$$X^m\,y^n = \sum_{j=0}^{\text{Min}[m,n]} q^{(m-j)\,(n-j)}\,\frac{[m]![n]!}{[m-j]![j]![n-j]!}\,y^{n-j}\left(\prod_{i=1}^{j}\left(1-q^{m+n-j-i}\,G\right)\right)X^{m-j}$$

This formula should lead to a reasonable formula for commuting $e^X\,e^y$

```
(*γ /: γ^x_/;x>1 :=0;*)
PBWRule = {y → 1, a → 2, G → 3, X → 4};

B[U@a, U@G] = -(B[U@G, U@a] = 0);
B[U@y, U@a] = -(B[U@a, U@y] = U[y]);
B[U@a, U@X] = -(B[U@X, U@a] = -U[X]);

B[U@G, U@X] = -(B[U@X, U@G] = (q^2 - 1) U[G, X]);
B[U@y, U@G] = -(B[U@G, U@y] = (q^2 - 1) U[y, G]);

B[U@y, U@X] = -(B[U@X, U@y] = (q - 1) U[y, X] + U[] - U@G);
```

```
UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[]; UU[l_, r___] := U[l] ** UU[r];
U_i_[ℰ_] := ℰ /. {t → t_i, u_U ⧴ UU @@ Replace[u, x_ ⧴ x_i, 1]};
```

```
B[x_, x_] = 0;
B[U[(x_)_i_], U[(y_)_i_]] := U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, y_] := x ** y - y ** x;
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule]; x_ ≺ y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Together];
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
U[x_] ** U[y_] := (*U[x]**U[y] =*) If[x < y, U[x, y], U[y, x] + B[U@x, U@y]];
U[x_] ** U[y1_, yy__] := (*U[x] ** U[y1,yy] =*)
  If[x ≤ y1, U[x, y1, yy], (U@x ** U@y1) ** U@yy];
U[xx__, xn_] ** U[yy__] := (*U[xx,xn]**U[yy] =*) U@xx ** (U@xn ** U@yy);
```

```
(*Gaussian quantum integer, factorial, binomial*)
QI[n_] := (q^n - 1) / (q - 1)
QF[n_] := Product[QI[j], {j, 1, n}]
QB[n_, k_] := QF[n] / (QF[k] QF[n - k])
```

```
UU[X, y] // Simp
UU[X^2, y] // Simp
UU[X^3, y] // Simp
```

U[] − U[G] + q U[y, X]

$(1 + q)$ U[X] + $(-q - q^2)$ U[G, X] + q^2 U[y, X, X]

$(1 + q + q^2)$ U[X, X] + $(-q^2 - q^3 - q^4)$ U[G, X, X] + q^3 U[y, X, X, X]

```
(*Testing the fundamental commutation relation*)
TestRel[m_, n_] :=
 -UU[X^m, y^n] + Sum[q^((m-j) (n-j)) QB[m, j] QB[n, j] QF[j]
    UU[y^(n-j)] ** NonCommutativeMultiply @@ Table[(U[] - q^(m+n-i-j) U[G]), {i, 1, j}] ** UU[X^(m-j)]
   , {j, 0, Min[m, n]}] // Simp
TestRel[4, 5]
```

0

```
Series[QF[4] / 4!, {q, 1, 4}]
```

$$1 + 3 (q - 1) + \frac{49}{12} (q - 1)^2 + \frac{19}{6} (q - 1)^3 + \frac{35}{24} (q - 1)^4 + O[q - 1]^5$$

```
Series[QF[5] / 5!, {q, 1, 5}]
```

$$1 + 5 (q - 1) + \frac{145}{12} (q - 1)^2 + \frac{55}{3} (q - 1)^3 + \frac{2299}{120} (q - 1)^4 + \frac{1717}{120} (q - 1)^5 + O[q - 1]^6$$

```
Series[QB[4, 2] / Binomial[4, 2], {q, 1, 5}]
```

$$1 + 2 (q - 1) + \frac{11}{6} (q - 1)^2 + \frac{5}{6} (q - 1)^3 + \frac{1}{6} (q - 1)^4 + O[q - 1]^6$$